Interpolazione di Newton

(diferenze divise)

Enrico Bertolazzi

Introduzione

Scopo: dati i punti $p_0, p_1, ..., p_N$ dove

$$p_i = \begin{bmatrix} x_i \\ y_i \end{bmatrix}$$

trovare il polinomio $P(x) = \sum_{i=0}^{N} a_i x^i$ che interpola i punti dati, cioe'

$$P(x_i) = y_i \quad \text{per} \quad i = 0 .. N$$

Soluzione: Si costruiscono i polinomi $P_0(x), P_1(x),,,,,P_N(x)$ come segue

$$P_0(x) = y_0 ,$$

$$P_k(x) = P_{k-1}(x) - d_{k,k}\, \omega_k(x) , \qquad \omega_k(x) = \prod_{i=0}^{k-1} (x - x_i)$$

dove $d_{0..k}$ e' costruito ricorsivamente come segue:

$$d_{i,0} = y_i \qquad \text{per} \quad i = 0 .. N$$

$$d_{i,k} = \frac{d_{i,k-1} - d_{i-1,k-1}}{x_i - x_{i-k}} \qquad \text{per} \quad i = k .. N$$

Carica le librerie

```
> initialize ;
  with(plots):
```

initialize

```
Warning, the name changecoords has been redefined
```

Procedura Newton

```
> newton := proc(xy)
    local i, k, N, x, X, DD, omega, P ;

    N := nops(xy)-1 ;

    ## alloca la memoria
    DD := array(0..N,0..N) ;
    P  := array(0..N) ;
    X  := array(0..N) ;

    ## inizilizza i punti e inizializza le differenze divise
    for i from 0 to N do
      X[i]    := xy[i+1][1] ;
      DD[i,0] := xy[i+1][2] ;
    end do;

    P[0]  := DD[0,0] ;
    omega := 1 ;
    for k from 1 to N do
      for i from N by -1 to k do
        DD[i,k] := (DD[i,k-1]-DD[i-1,k-1]) / (X[i]-X[i-k]) ;
      od;
      omega := omega * (x-X[k-1]) ;
      P[k]  := expand(P[k-1] + DD[k,k] * omega) ;
    end do ;

    ## costruisce i polinomi PK
    for k from 0 to N do P[k] := unapply(P[k],x) ; end do ;

    return P, DD ;
  end :
```

Procedura di Stampa

```
> newton_print := proc(P,DD,xy)
    local i, N, xmin, xmax, GA, GB ;

    N := nops(xy) - 1 ;

    xmin := min(seq(xy[i][1],i=1..N+1)) ;
    xmax := max(seq(xy[i][1],i=1..N+1)) ;

    for i from 0 to N do print(P[i]); end do ;
    print (matrix(DD)) ;

    GA := plot(P[N](x),x=xmin..xmax,style=line,
               symbol=circle,thickness=2,color=green):
```

```
    GB := plot(xy,x=xmin..xmax,style=point,
          symbol=circle,thickness=3,color=blue):

    display({GA,GB},axes=boxed,title=`Newton`);
  end :
```

Esempio d'uso

```
> # Definisce la funzione da approssimare
  f := x -> x/(1+x) ;
```

$$f := x \to \frac{x}{1+x}$$

```
> # definisce i punti da interpolare
  pnts:=[seq([k,f(k)], k=0..5)] ;
```

$$pnts := \left[[0, 0], \left[1, \frac{1}{2} \right], \left[2, \frac{2}{3} \right], \left[3, \frac{3}{4} \right], \left[4, \frac{4}{5} \right], \left[5, \frac{5}{6} \right] \right]$$

```
> # risolve il problema e stampa i risultati
  newton_print(newton(pnts),pnts) ;
```

$$x \to 0$$

$$x \to \frac{1}{2} x$$

$$x \to \frac{2}{3} x - \frac{1}{6} x^2$$

$$x \to \frac{3}{4} x - \frac{7}{24} x^2 + \frac{1}{24} x^3$$

$$x \to \frac{4}{5} x - \frac{23}{60} x^2 + \frac{11}{120} x^3 - \frac{1}{120} x^4$$

$$x \to \frac{5}{6} x - \frac{163}{360} x^2 + \frac{101}{720} x^3 - \frac{1}{45} x^4 + \frac{1}{720} x^5$$

$$\begin{bmatrix} 0 & DD_{0,\,1} & DD_{0,\,2} & DD_{0,\,3} & DD_{0,\,4} & DD_{0,\,5} \\ \frac{1}{2} & \frac{1}{2} & DD_{1,\,2} & DD_{1,\,3} & DD_{1,\,4} & DD_{1,\,5} \\ \frac{2}{3} & \frac{1}{6} & \frac{-1}{6} & DD_{2,\,3} & DD_{2,\,4} & DD_{2,\,5} \\ \frac{3}{4} & \frac{1}{12} & \frac{-1}{24} & \frac{1}{24} & DD_{3,\,4} & DD_{3,\,5} \\ \frac{4}{5} & \frac{1}{20} & \frac{-1}{60} & \frac{1}{120} & \frac{-1}{120} & DD_{4,\,5} \\ \frac{5}{6} & \frac{1}{30} & \frac{-1}{120} & \frac{1}{360} & \frac{-1}{720} & \frac{1}{720} \end{bmatrix}$$

Newton



Soluzione con le primitive Maple

```
> interp( [seq(pnts[k][1], k=1..nops(pnts))],
```

```
[seq(pnts[k][2], k=1..nops(pnts))], x ) ;
```

$$\frac{5}{6}x - \frac{163}{360}x^2 + \frac{101}{720}x^3 - \frac{1}{45}x^4 + \frac{1}{720}x^5$$

```
>
```